

Qt + CMake + vcpkg

How to build a static Qt Quick Controls 2 app using CMake and vcpkg


05.11.2019

alcroito@Qt Company




Starting point

- › Qt Quick Controls 2 example “texteditor”
- › Mix of C++ and QML
- › qmake project



Goals

- › Build app using CMake
- › Build it statically
- › Use a third party library from vcpkg
- › Target multiple platforms



Let's dive in

.pro file

```
1. TEMPLATE = app
2. TARGET = texteditor
3. QT += quick quickcontrols2
4. qtHaveModule(widgets): QT += widgets

5. cross_compile: DEFINES += QT_EXTRA_FILE_SELECTOR="touch"

6. HEADERS += \
7.     documenthandler.h

8. SOURCES += \
9.     texteditor.cpp \
10.    documenthandler.cpp

11. RESOURCES += \
12.    texteditor.qrc
```

Initial CMakeLists.txt

```
1. cmake_minimum_required(VERSION 3.15)
2. project(texteditor LANGUAGES CXX)
3. find_package(Qt5 5.14 REQUIRED COMPONENTS QuickControls2)
4. find_package(Qt5 5.14 COMPONENTS Widgets)
5. set(CMAKE_AUTOMOC ON)
6. set(CMAKE_AUTORCC ON)
7. add_executable(texteditor texteditor.cpp documenthandler.cpp texteditor.qrc)
8. if(CMAKE_CROSSCOMPILING)
9.     target_compile_definitions(texteditor PRIVATE QT_EXTRA_FILE_SELECTOR=\"touch\")
10.endif()
11.target_link_libraries(texteditor PRIVATE Qt5::Quick Qt5::QuickControls2)
12.if(TARGET Qt5::Widgets)
13.    target_link_libraries(texteditor PRIVATE Qt5::Widgets)
14.endif()
```

Minimum CMake version
to build project

Find Qt libraries

Enable auto moc and rcc

Create executable

Mobile UI support

Use Qt libraries

Static builds?

Before Qt 5.14

```
1. set(build_static TRUE)
2. if(build_static)
3.     # QuickTemplates2 package is also explicitly required for some reason.
4.     find_package(Qt5QuickTemplates2 REQUIRED)
5.     target_link_libraries(texteditor PRIVATE Qt5::QuickTemplates2)
6.
7.     # Find all system libs
8.     find_library(FWSecurity Security)
9.     find_library(FWSystemConfiguration SystemConfiguration)
10.    find_library(FWCoreText CoreText)
11.    find_library(FWNetwork Network)
12.    find_library(FWCoreFoundation CoreFoundation)
13.    find_library(FWFoundation Foundation)
14.    find_library(FWCoreGraphics CoreGraphics)
15.    find_library(FWCoreServices CoreServices)
16.    find_library(FWAppKit AppKit)
17.    find_library(FWCoreVideo CoreVideo)
18.    find_library(FWIOKit IOKit)
19.    find_library(FWIOSurface IOSurface)
20.    find_library(FWCarbon Carbon)
21.    find_library(FWMetal Metal)
22.    find_library(FWQuartzCore QuartzCore)
23.
24.    find_library(CUPS_LIBRARIES NAMES cups )
25.    find_package(OpenGL)
26.    find_package(ZLIB)
27.
28.    # Link against qt provided 3rd party libs
29.    target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libqtharfbuzz.a)
30.    target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libqtlibpng.a)
```

```
1.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libqtpcre2.a)
2.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libqtfreetype.a)
3.
4.     # Link against the various platform support plugins
5.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libQt5FontDatabaseSupport.a)
6.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libQt5PrintSupport.a)
7.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libQt5GraphicsSupport.a)
8.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libQt5ClipboardSupport.a)
9.     target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libQt5ThemeSupport.a)
10.    target_link_libraries(texteditor PRIVATE ${qt_dir}/lib/libQt5AccessibilitySupport.a)
11.
12.    # Link against the system libs
13.    target_link_libraries(texteditor PRIVATE ${FWSecurity} ${FWSystemConfiguration} ${FWCoreText}
14.    ${FWNetwork} ${FWCoreFoundation} ${FWCoreGraphics})
15.    target_link_libraries(texteditor PRIVATE ${FWIOKit} ${FWIOSurface} ${FWCarbon} ${FWMetal}
16.    ${FWQuartzCore})
17.    target_link_libraries(texteditor PRIVATE ${FWFoundation} ${FWCoreServices} ${FWAppKit}
18.    ${FWCoreVideo} OpenGL::GL ZLIB::ZLIB ${CUPS_LIBRARIES})
19.
20.    # Link against the QPA plugin
21.    target_link_libraries(texteditor PRIVATE Qt5::QCocoaIntegrationPlugin)
22.
23.    # Link against the Qt Quick plugins
24.    set(qml_plugin_dir "${qt_dir}/qml")
25.    target_link_libraries(texteditor PRIVATE ${qml_plugin_dir}/QtQuick.2/libqtquick2plugin.a)
26.    target_link_libraries(texteditor PRIVATE ${qml_plugin_dir}/QtQuick.Window.2/libwindowplugin.a)
27.    target_link_libraries(texteditor PRIVATE
28.    ${qml_plugin_dir}/QtQuick.Controls.2/libqtquickcontrols2plugin.a)
29.    target_link_libraries(texteditor PRIVATE
30.    ${qml_plugin_dir}/Qt/labs/platform/libqtlabsplatformplugin.a)
31.    target_link_libraries(texteditor PRIVATE
32.    ${qml_plugin_dir}/QtQuick.Templates.2/libqtquicktemplates2plugin.a)
33.
34.    # Initialize the plugins
35.    target_sources(texteditor PRIVATE init_plugins.cpp)
36.    endif()
```


After Qt 5.14+

```
1. cmake_minimum_required(VERSION 3.15)
2. project(texteditor LANGUAGES CXX)
3. find_package(Qt5 5.14 REQUIRED COMPONENTS QuickControls2 QmlImportScanner)
4. find_package(Qt5 5.14 COMPONENTS Widgets)
5. set(CMAKE_AUTOMOC ON)
6. set(CMAKE_AUTORCC ON)
7. add_executable(texteditor texteditor.cpp documenthandler.cpp texteditor.qrc)
8. if(CMAKE_CROSSCOMPILING)
9.     target_compile_definitions(texteditor PRIVATE QT_EXTRA_FILE_SELECTOR=\"touch\")
10.endif()
11.target_link_libraries(texteditor PRIVATE Qt5::Quick Qt5::QuickControls2)
12.if(TARGET Qt5::Widgets)
13.    target_link_libraries(texteditor PRIVATE Qt5::Widgets)
14.endif()
15.# Import qml plugins for static builds.
16.qt5_import_qml_plugins(texteditor)
```

Regular plugins found automatically


Find qml import scanner

Import / build qml plugins

Third party libraries?

Third party libraries

- › What's a useful feature for a text editor?
- › Encrypted documents!
- › Which C++ library provides encryption?
- › Botan!



Botan

- › BSD-licensed cryptographic library
- › Written in C++
- › Supports desktop and mobile platforms
- › Where to get it?

vcpkg

- › Contains recipes for building many libraries from source
- › Targets multiple platforms (Windows / macOS / Linux) (in vcpkg lingo – triplets)
- › Many packages integrate well with CMake
- › Recipes written in CMake
- › Incidentally, provides a recipe for Botan

Building Botan

```
1.cd <vcpkg_dir>  
2.export VCPKG_DEFAULT_TRIPLET=x64-osx  
3../vcpkg install botan
```

Find Botan?

```
1. # FindBotan.cmake

2. find_path(BOTAN_INCLUDE_DIRS NAMES botan/botan.h
3.           DOC "The botan include directory")

4. find_library(BOTAN_LIBRARIES NAMES botan botan-2
5.             DOC "The botan library")

6. include(FindPackageHandleStandardArgs)
7. find_package_handle_standard_args(Botan REQUIRED_VARS BOTAN_LIBRARIES BOTAN_INCLUDE_DIRS)

8. if(Botan_FOUND)
9.     add_library(Botan::Botan UNKNOWN IMPORTED)

10.    set_target_properties(Botan::Botan PROPERTIES
11.                          IMPORTED_LOCATION "${BOTAN_LIBRARIES}"
12.                          INTERFACE_INCLUDE_DIRECTORIES "${BOTAN_INCLUDE_DIRS}")
13.endif()

14.mark_as_advanced(BOTAN_LIBRARIES BOTAN_INCLUDE_DIRS)
```

Find headers

Find static libraries

Create CMake target

Assign location of headers
and libraries to target

Use Botan!

```
1. cmake_minimum_required(VERSION 3.15)
2. include(cmake/app_utils.cmake)
3. setup_vcpkg_before_project()
4. project(encrypted_texteditor LANGUAGES CXX)

5. find_package(Qt5 5.14 REQUIRED COMPONENTS QuickControls2 QmlImportScanner)
6. find_package(Qt5 5.14 COMPONENTS Widgets)

7. list(APPEND CMAKE_MODULE_PATH "${CMAKE_CURRENT_SOURCE_DIR}/cmake")
8. find_package(Botan REQUIRED)

9. set(CMAKE_AUTOMOC ON)
10. set(CMAKE_AUTORCC ON)

11. add_executable(texteditor texteditor.cpp documenthandler.cpp texteditor.qrc)

12. if(CMAKE_CROSSCOMPILING)
13.     target_compile_definitions(texteditor PRIVATE QT_EXTRA_FILE_SELECTOR=\"touch\")
14. endif()

15. target_link_libraries(texteditor PRIVATE Qt5::Quick Qt5::QuickControls2 Botan::Botan)
16. if(TARGET Qt5::Widgets)
17.     target_link_libraries(texteditor PRIVATE Qt5::Widgets)
18. endif()

19. # Import qml plugins for static builds.
20. qt5_import_qml_plugins(texteditor)
```

Set up vcpkg

Find Botan

Use Botan

Set up vcpkg

```
1. function(setup_vcpkg_before_project)
2.     if(DEFINED ENV{VCPKG_ROOT})
3.         set(vcpkg_toolchain_path "$ENV{VCPKG_ROOT}/scripts/buildsystems/vcpkg.cmake")
4.         set(CMAKE_TOOLCHAIN_FILE "${vcpkg_toolchain_path}" CACHE STRING "" FORCE)
5.     endif()





6.     if(DEFINED ENV{VCPKG_DEFAULT_TRIPLET} AND NOT VCPKG_TARGET_TRIPLET)
7.         set(VCPKG_TARGET_TRIPLET "$ENV{VCPKG_DEFAULT_TRIPLET}" CACHE STRING "")
8.     endif()
9. endfunction()
```

Set vcpkg toolchain


Set vcpkg triplet

Multiple platforms?

Platforms



iPhone Xr — 12.2



Wait. Mobile too?

Mobile support

- › No upstream support for mobile platforms in vcpkg
 - › My fork contains modifications to build Botan for mobile (due to Botan not using CMake as its build system)
- › Setting up CMake projects for iOS and Android is a trial and error process
 - › No single source of truth for documentation
- › Still possible!

iOS specifics

Certain things need to be done before the first *project()* call

```
1. set(MACOSX_BUNDLE_BUNDLE_NAME "CryptEd" PARENT_SCOPE)
2. set(MACOSX_BUNDLE_GUI_IDENTIFIER "io.qt.alcroito.crypted" PARENT_SCOPE)
3. set(MACOSX_BUNDLE_INFO_STRING "CryptEd" PARENT_SCOPE)
4. set(MACOSX_BUNDLE_LONG_VERSION_STRING "0.1.0" PARENT_SCOPE)
5. set(MACOSX_BUNDLE_SHORT_VERSION_STRING "0.1" PARENT_SCOPE)
6. set(MACOSX_BUNDLE_BUNDLE_VERSION "0.1" PARENT_SCOPE)
7. set_target_properties(${target} PROPERTIES MACOSX_BUNDLE_INFO_PLIST
8.     "${CMAKE_CURRENT_SOURCE_DIR}/Info_ios.plist.in")

9. if(IOS)
10.     target_link_options(texteditor PRIVATE "-Wl,-e,_qt_main_wrapper")
11.endif()

12.set(CMAKE_OSX_DEPLOYMENT_TARGET "12.2" CACHE STRING "")
13.set(CMAKE_XCODE_ATTRIBUTE_CODE_SIGN_IDENTITY "iPhone Developer")

14.# Add the team identifier here, otherwise code-signing for the device won't work.
15.# Can also be set via CMake cache variable.
16.set(CMAKE_XCODE_ATTRIBUTE_DEVELOPMENT_TEAM "")

17.set(CMAKE_OSX_SYSROOT "iphoneos")
18.set(CMAKE_OSX_ARCHITECTURES "arm64" CACHE STRING "")
```

Specify usual Info.plist keys

Need to manually change application entry point

Set up code signing identity, development team, deployment target

Specify architecture and device or simulator

Android specifics

No static builds : (

```
1. if(ANDROID)
2.     add_library("${target}" MODULE ${ARGN})
3.     # Not having this flag set might cause the executable to have main()
4.     # hidden and it can thus no longer be loaded through dlopen().
5.     set_property(TARGET "${target}" PROPERTY C_VISIBILITY_PRESET default)
6.     set_property(TARGET "${target}" PROPERTY CXX_VISIBILITY_PRESET default)
7. else()
8.     add_executable("${target}" WIN32 MACOSX_BUNDLE ${ARGN})
9. endif()
```

add_library(MODULE)
instead of *add_executable*

```
10. if(DEFINED CMAKE_TOOLCHAIN_FILE AND vcpkg_toolchain_path)
11.     get_filename_component(supplied_toolchain_file "${CMAKE_TOOLCHAIN_FILE}" ABSOLUTE)
12.     if(NOT supplied_toolchain_file STREQUAL vcpkg_toolchain_path)
13.         set(VCPKG_CHAINLOAD_TOOLCHAIN_FILE "${CMAKE_TOOLCHAIN_FILE}" CACHE STRING "")
14.     endif()
15. endif()
```

Using vcpkg requires
chainloading a toolchain

```
1. cd android_build
2. QT_PATH=/Users/alex/qt/qt514_android/
3. ANDROID_NDK_HOME=/Users/alex/android/sdk/ndk/20.0.5594570
4. ANDROID_SDK_HOME=/Users/alex/android/sdk
5. cmake .. -G"Unix Makefiles" -DANDROID_ABI=arm64-v8a \
6.     -DANDROID_SDK=$ANDROID_SDK_HOME
7.     -DANDROID_NDK=$ANDROID_NDK_HOME \
8.     -DANDROID_NATIVE_API_LEVEL=28 \
9.     -DCMAKE_TOOLCHAIN_FILE=/Users/alex/vcpkg/scripts/buildsystems/vcpkg.cmake \
10.    -DVCPKG_TARGET_TRIPLET=arm64-android \
11.    -DCMAKE_PREFIX_PATH=$QT_PATH -DCMAKE_FIND_ROOT_PATH=$QT_PATH \
12.    -DVCPKG_CHAINLOAD_TOOLCHAIN_FILE=$ANDROID_NDK_HOME/build/cmake/android.toolchain.cmake
13. make -j16
14. make -j16 VERBOSE=1 apk # Calls androiddeployqt
```

Need to specify many
params when building

Minimum required versions

- › Qt 5.14+
- › CMake 3.15+ for iOS (lower for other platforms)
- › Qt Creator 4.11+ for Android CMake integration (iOS CMake integration currently missing)

Links

- › Full project source – https://github.com/alcroito/qt_world_summit_2019_cmake_vcpkg_app
https://tiny.cc/qt_vcpkg_app
- › Vcpkg fork for mobile support – https://github.com/alcroito/vcpkg/tree/qt_world_summit_2019_botan
- › Slides included in Repo



Thanks. Questions?